pluginManagement {
  repositories {
    mavenCentral()
    gradlePluginPortal()
  }
}
plugins {
  // https://github.com/diffplug/blowdryer/blob/main/CHANGELOG.md
  id 'com.diffplug.blowdryerSetup' version '1.7.1'
  // https://github.com/diffplug/spotless/blob/main/plugin-gradle/CHANGES.md
  id 'com.diffplug.spotless' version '7.2.1' apply false
  // https://github.com/diffplug/spotless-changelog/blob/main/CHANGELOG.md
  id 'com.diffplug.spotless-changelog' version '3.1.2' apply false
  // https://plugins.gradle.org/plugin/com.gradle.develocity
  id 'com.gradle.develocity' version '4.1.1'
  // https://plugins.gradle.org/plugin/com.gradle.plugin-publish
  id 'com.gradle.plugin-publish' version '2.0.0' apply false
  // https://github.com/equodev/equo-ide/blob/main/plugin-gradle/CHANGELOG.md
  id 'dev.equo.ide' version '1.7.8' apply false
  // https://github.com/gradle-nexus/publish-plugin/releases
  id 'io.github.gradle-nexus.publish-plugin' version '2.0.0' apply false
  // https://plugins.gradle.org/plugin/org.jetbrains.dokka
  id 'org.jetbrains.dokka'                       version '2.0.0' apply false
  // https://plugins.gradle.org/plugin/org.jetbrains.kotlin.jvm
  id 'org.jetbrains.kotlin.jvm'                  version '2.2.20' apply false
  // https://plugins.gradle.org/plugin/org.jetbrains.kotlin.plugin.serialization
  id 'org.jetbrains.kotlin.plugin.serialization' version '2.2.20' apply false
  // https://plugins.gradle.org/plugin/org.jetbrains.kotlin.multiplatform
  id 'org.jetbrains.kotlin.multiplatform' version '2.2.20' apply false
  // https://github.com/adamko-dev/dokkatoo/releases
  id 'dev.adamko.dokkatoo-html' version '2.4.0' apply false
}

blowdryerSetup {
  github 'diffplug/blowdryer-diffplug', 'tag', '9.1.0'
  //devLocal '../blowdryer-diffplug'
}

// Core library
include 'selfie-lib'
// JUnit 4 & 5
include 'selfie-runner-junit5'
include 'undertest-junit5'
include 'undertest-junit5-kotest'
include 'undertest-junit-vintage'
// Kotest
include 'selfie-runner-kotest'
include 'undertest-kotest'

// examples for docs and whatsuch
if (JavaVersion.current() >= JavaVersion.VERSION_15) {
  include 'example-junit5'
}
include 'example-kotest'

rootProject.name = 'selfie'

apply from: com.diffplug.blowdryer.Blowdryer.file('base/buildscan-on-ci.gradle')